	Functions:
 

 



Function KeyPressed: Boolean;
 
Determines if a key on the keyboard has been pressed. Keypressed returns true if a key on the keyboard has been pressed, false otherwise. Read the key with ReadKey 
 




Function ReadKey: Char;
 
Reads a key from the keyboard. The char read is not echoed to the screen. If KeyPressed was true before the ReadKey function was called, the char is returned immediatly, otherwise the function will return before a key is typed.
 
Some of the special keys on the keyboard generate extended scan-codes (That means, that ReadKey will first return a #0 char, and after that, the extended code).
 
For a list over extended key codes that can be read with ReadKey take a look at the table here.
 
Reading of extended key codes was not supported in the original WinCrt unit. 
 




Function ReadBuf(Buffer: PChar; Count: Word): Word;
 
Reads a line from the CRT window. Buffer points to a line buffer that has room for up to Count chars. Up to Count-2 chars can be entered, and an end-of-line marker (#13#10) will be appended to the line when the user presses enter.
 
If CheckEOF is True, the input can be terminted with Ctrl+Z as well, and the line will have a end-of-file marker (#26) appended.
 
The return value is the number of chars entered, including the end-of-file or end-of-line marker. 
 
A small example: 

                  program testprog;

                  uses

                    Win32Crt;

                  Var 

                    ReadBuffer : Array [0..81] of char;

                  begin

                    CheckEOF:=True;

                    Writeln(ReadBuf(ReadBuffer,81));

                    Writeln(ReadBuffer);

                  end.




Function WhereX: Integer;
 
Returns the X-coordinate of the current cursor location. The returned value is 1-based, and is the same as Cursor.X + 1. 
 




Function WhereY: Integer;
 
Returns the Y-coordinate of the current cursor location. The returned value is 1-based, and is the same as Cursor.Y + 1. 
 




Function SmartInput(X,Y:Integer;Len:Integer;Var Value:AnsiString;OkKeys : KeySet;TabAdvance:Boolean;AutoAdvance:Boolean):Integer;
 
This function reads a string from the keyboard, and return the key used to exit the procedure, as in a normal ReadKey where this is the second char returned (the first was #0) (So the char returned is the extended char eg. for PgUp, PgDn, Up and down).
 
The reading will start at screen postion X,Y, and will allow up to Len chars to be read
 
If TabAdvance is true, then a tab-char will return the same char as Down, and Shift+Tab will return the same code as Up.
 
If Autoadvance is true, then if the char just entered was the last that could fit into the string, then the function returns the same code as Down
 
The string read is returned in Value.
 
Only chars in OKKeys will be allowed in the string.
 
A small example: 

                program testprog;

                uses

                  Win32Crt;

                Var

                  ReadBuffer : String;

                begin

                  GotoXY(10,10);

                  ReadBuffer:='-no name -';

                  Write('Input your name:');

                  SmartInput(28,10,25,ReadBuffer,['A'..'Z','a'..'z',' '],False,False);

                  GotoXY(1,20);

                  Writeln(ReadBuffer);

                end.




Function ScreenInput(Var TheJob:Array of InputJobs):boolean;
 
A function, that can be used to input a complete screen-full of information in one go. You make an array of InputJobs, that is filled with the wanted information, then calls this function to get the data read from the screen. If it returns False, then user exited by pressing ESC, else ENTER
 
A small example: 

                program Inputtest;

                uses

                  Win32CRT;

                Var

                  TestB,TestA : Array[0..9] of InputJobs;

                  Taller:Integer;

                Begin

                  ShowScroll:=False;    { Don't show scrollbars }

                  UseScrollKeys:=False; { Make sure we have control over the scrollkeys }

                  ScrollScreen:=False;  { Don't scroll after last line }

                  CanResize:=False;     { Don't resize the window }

                  ClrScr;               { Clear the screen }

                  TextOutPos(10,2,'This is a small TEST-PROGRAM for ZieglerSoft''s Win32CRT module');

                  TextOutPos(7,24,'Exit it with ''ESC'' or by going to the last field and pressing ''Enter''');

                  FillScreenChar(1,1,80,'*'); { Draw a row of starts }

                  FillScreenChar(1,3,80,'*'); { Draw a row of starts }

                  FillScreenChar(1,23,80,'*'); { Draw a row of starts }

                  FillScreenChar(1,25,80,'*'); { Draw a row of starts }

                  ScrollScreen:=True;  { It is OK to scroll after last line again }

                  { Put data into the input-screen variables }

                  With TestA[0] Do begin

                    X:=15; Y:=5; L:=10; S:='1100';

                    AK:=['0'..'9','.',','];

                    X1:=3; Y1:=5; P:=' A number: ';

                  End;

                  With TestA[1] Do begin

                    X:=41; Y:=5; L:=30; S:='A text';

                    X1:=31; Y1:=5; P:=' A text: ';

                  End;

                  With TestA[2] Do begin

                    X:=27; Y:=9; L:=15; S:='';

                    X1:=3; Y1:=9; P:=' A bunch of textlines: ';

                  End;

                  With TestA[3] Do begin

                    X:=27; Y:=10; L:=15; S:='more text';

                  End;

                  With TestA[4] Do begin

                    X:=27; Y:=11; L:=15; S:='more text';

                  End;

                  With TestA[5] Do begin

                    X:=27; Y:=12; L:=15; S:='and more';

                  End;

                  With TestA[6] Do begin

                    X:=13; Y:=13; L:=15; S:='and more';

                  End;

                  With TestA[7] Do begin

                    X:=13; Y:=14; L:=15; S:='and more';

                  End;

                  With TestA[8] Do begin

                    X:=13; Y:=15; L:=15; S:='even more';

                  End;

                  With TestA[9] Do begin

                    X:=13; Y:=16; L:=15; S:='text';

                  End;

                  TestB:=TestA; { Save a copy of the inputs }

                  { Show the screen, and return the result }

                  If ScreenInput(TestA) then Begin

                    ClrScr;

                    Writeln('You exited by pressing ''Enter'' so this is the input:');

                    Writeln;

                    For Taller:=0 to High(TestA) Do

                    Writeln('Nr.: ',Taller,' Text: ',TestA[Taller].S);

                  End Else begin

                    ClrScr;

                    Writeln('You exited by pressing ''ESC'' so you keep the stuff:');

                    Writeln;

                    For Taller:=0 to High(TestB) Do

                    Writeln('Nr.: ',Taller,' Text: ',TestB[Taller].S);

                  End;

                  Writeln;

                  Writeln('To end the program, press "anykey"');

                  ReadKey;

                  DoneWinCrt;

                end.




 
Procedures:
 

 
Procedure InitWinCrt;
 
Creates the CRT window (If it has not already been created). A Read, ReadLn, Write or WriteLn automatically calls this function to ensure that the CRT window exists.
 
InitWinCrt uses WindowOrg, WindowSize, ScreenSize, WindowTitle and ShowScroll to determnie the characteristics of the CRT window. 
 




Procedure DoneWinCrt;
 
Destroys the CRT window if it has not already been detroyed. Calling DoneWinCrt just before the program ends prevents the CRT window from entering the inactive state, and therefore, the user is not required to close the window 
 




Procedure WriteBuf(Buffer: PChar; Count: Word);
 
Writes a block of chars to the CRT window. Buffer points to the first char int the block and count contains the number of chars to write.
 
If Autotracking is True, the CRT window is scrolled if needed to ensure that the cursor is visible after a block of chars is written. 
 




Procedure WriteChar(Ch: Char);
 
Writes a single char to the CRT window at the current cursorposition. 
 




Procedure GotoXY(X, Y: Integer);
 
Moves the cursor to the given coordinates whitin the virtual screen. The upper left corner corresponds to ()1,1). The Cursor variable is set to (X-1,Y-1), because it stores the cursor position relative to (0,0) instead of relative to (1,1) 
 




Procedure ClrScr;
 
Clears the active window and returns the Cursor to the upper left-hand corner. 
 




Procedure ClrEol;
 
Clears all chars from the cursor, to the end of the line, without moving the cursor. 
 




Procedure CursorTo(X, Y: Integer);
 
Moves the Cursor to the given coordinates within the virtual screen. The upper left corner corrosponds to (0,0). Set the Cursor to (X,Y). 
 




Procedure ScrollTo(X, Y: Integer);
 
Scrolls the CRT window to show the virtual screen location given by (X,Y) in the upper left corner. The location described with coordinates (0,0) corresponds to the upper left corner of the virtual screen. 
 




Procedure TrackCursor;
 
Scrolls the CRT window to ensure the cursor is visible. 
 




Procedure AssignCrt(var F: Text);
 
Associates a text file with the CRT.
 
AssignCrt works like the normal Assign procedure, except that no filename is given. Instead, the textfile is associated with the CRT window.
 
All Read, ReadLn, Write and WriteLn will then go to the CRT window when done on the textfile. 
 




Procedure Delay(Msec:Word);
 
Dealys for a specified number of milliseconds (aprox). 
 




Procedure TextOutPos(X,Y:Integer;What:String);
 
Writes out the string What at the position (X,Y), without moving the cursor. 
 




Procedure FillScreenChar(X,Y:Integer;Length:Integer;C:Char);
 
Writes out a string of Char with the length Length at position (X,Y), without moving the cursor. 
 




Procedure Read(V1 [,V2,...,VN]);
 
Reads variables from the screen. 
 




Procedure ReadLn(V1 [,V2,...,VN]);
 
Does a Read and then skips to the next line. 
 




Procedure Write(P1 [,P2,...,PN]);
 
Writes out the P paramters to the virtual screen 
 




Procedure WriteLn(P1 [,P2,...,PN]);
 
Does a Write and then outputs an end-of-line marker. 
 




 
Types:
 

 
Type Keyset = Set of Char;
 
Used by SmartInput and ScreenInput 
 




Type InputJobs = Record
    X : Integer;
    Y : Integer;
    X1: Integer;
    Y1: Integer;
    L : Integer;
    S : AnsiString;
    P : AnsiString;
    AK: keySet;
End;
 
Used by ScreenInput to define every single input. 
 




 
Constant/Variables:
 

 
Var ScreenBuffer : PChar;
 
The current screen content. 
 




Var WindowOrg : TPoint = (X: cw_UseDefault; Y: cw_UseDefault);
 
Determines the inital location of the CRT window. The default location lets Windows select a suitable location for the CRT window. You can change the initial location by assigning new values to the X and Y values before the CRT window is created. 
 




Var WindowSize : TPoint = (X: cw_UseDefault; Y: cw_UseDefault);
 
Determines the initial size of the CRT window. The default size lets Windows select a suitable size for the CRT window. You can change the initial size by assigning new values for the X and Y values before the CRT window is created. 
 




Var ScreenSize : TPoint = (X: 80; Y: 25);
 
Determines the width and height in chars of the virtual screen whitin the CRT window. The default screen size is 80 cols and 25 lines.
 
You can change the size of the virtual screen by assigning new values to X and Y before the CRT window is created. The value of ScreenSize.X X ScreenSize.Y can not exceed 65520. 
 




Var Cursor : TPoint = (X: 0; Y: 0);
 
A read-only varaiable that contains the current position of the cursor, within the virtual screen. The upper left corner corresponds to (0,0). 
 




Var Origin : TPoint = (X: 0; Y: 0);
 
Contains the virtual screencoordinate of the char-cell displayed in the upper left corner of the CRT window. This variable is read-only. 
 




Var AutoTracking : Boolean = True;
 
Enables/Disables the windows automatic scrolling to keep the cursor visible.
 
When AutoTracking is True, the CRT window automatically scrolls to ensure that the cursor is visible after each Write and WriteLn. If False, the CRT window will not scroll and the text written might not be visible to the user. 
 




Var CheckEOF : Boolean = False;
 
Enables/Disables the end-of-file char.
 
When CheckEOF is True, an end-of-file char is generated if you press Ctrl+Z while reading from a file assigned to the screen.
 
When False, pressing Ctrl+Z has no effect. 
 




Var CheckBreak : Boolean = True;
 
Enables/Disables the check for Ctrl+Break.
 
When True the user can terminte the application at any time, by pressing Ctrl+C, Ctrl+Break, Clicking the close command in the controlmenu, pressing Alt+F4 or by double-clicking the control-menu. This will turn the application into the inactive state.
 
When False, theese features are disbled. 
 




Var ShowScroll : Boolean = True;
 
Enables/Disables the scrollbars in the CRT window. Can only be used before the CRT window is created.
 
False will trun off the scrollbars, True will turn them on. 
 




Var ScrollScreen : Boolean = True;
 
Enables/Disables the automatic scrolling when the last char of the last coloumn is reached.
 
If False no scrolling will take place. When True a normal scroll takes place when the last line is filled. 
 




Var UseScrollKeys : Boolean = True;
 
Enables/Disables the use of the arrowkeys on the keyboard.
 
If True the arrowkeys will scroll the virtual screen around in the CRT window.
If False the arrowkeys can be used in the ReadKey function, and will not scroll the screen around. 
 




Var CanResize : Boolean = True;
 
Enables/Disables the normal resize-functions in Windows. If True, the user can resize the CRT window, if False, the user can't resize the CRT window. 
 




Var CrtWindow : HWnd = 0;
 
The handle of the CRT window (Only available when the CRT window is created). 
 




Var WindowTitle : Array[0..79] of Char;
 
The title of the CRT window when the application is active. Set it before the CRT window is created. 
 




Var InactiveTitle: PChar='Inactive program: "%s" ';
 
The title of the CRT window when the application is inactive. 
 




Var AllKeys : KeySet = [#000..#255];
 
A constant, that defines every key-code that can be received from the keyboard. Used by SmartInput and ScreenInput 
 




 
Tables: 


	Codes returned from The ReadKey function
 
When using ReadKey, the return value will be #0 if an extended key is used. Not all extended keys normally found in the old CRT unit is possible with Win32Crt, so here is a list of the codes supported
 

	Key
	Normal
	With Shift
	With Ctrl
	With Alt

	F1
	#0 + #59
	#0 + #84
	#0 + #94
	#0 + #104

	F2
	#0 + #60
	#0 + #85
	#0 + #95
	#0 + #105

	F3
	#0 + #61
	#0 + #86
	#0 + #96
	#0 + #106

	F4
	#0 + #62
	#0 + #87
	#0 + #97
	#0 + #107

	F5
	#0 + #63
	#0 + #88
	#0 + #98
	#0 + #108

	F6
	#0 + #64
	#0 + #89
	#0 + #99
	#0 + #109

	F7
	#0 + #65
	#0 + #90
	#0 + #100
	#0 + #110

	F8
	#0 + #66
	#0 + #91
	#0 + #101
	#0 + #111

	F9
	#0 + #67
	#0 + #92
	#0 + #102
	#0 + #112

	F10
	#0 + #68
	#0 + #93
	#0 + #103
	#0 + #113

	F11
	#0 + #133
	#0 + #135
	#0 + #137
	#0 + #139

	F12
	#0 + #134
	#0 + #136
	#0 + #138
	#0 + #140

	INS
	#0 + #82
	 
	#0 + #146
	#0 + #162

	DEL
	#0 + #83
	 
	#0 + #255
	#0 + #163

	TAB
	#9
	#0 + #15 + #9
	#0 + #148
	 

	
If the variable UseScrollKeys is set to false, the folowing codes are also supported
 

	Key
	Normal
	With Shift
	With Ctrl
	With Alt

	LEFT
	#0 + #75
	 
	#0 + #115
	#0 + #155

	RIGHT
	#0 + #77
	 
	#0 + #116
	#0 + #157

	UP
	#0 + #72
	 
	#0 + #141
	#0 + #152

	DOWN
	#0 + #80
	 
	#0 + #145
	#0 + #160

	HOME
	#0 + #71
	 
	#0 + #119
	#0 + #151

	END
	#0 + #79
	 
	#0 + #117
	#0 + #159

	PGUP
	#0 + #73
	 
	#0 + #132
	#0 + #153

	PGDN
	#0 + #81
	 
	#0 + #118
	#0 + #161

	
The normal keys on the keybord returns the normal letter/number, without any #0 in front. You can't return ALT + CHAR or ALT + NUMBER
 


